Solving rolling disk inside another using symbolic computation



By Nasser M. Abbasi

April 28, 2011

Introduction

The following shows how to use *Mathematica* symbolic computation to determine the Lagrangian of the above problem and solve numerically for the equations of the motion. The above shows a small disk which rotate inside a larger disk. The small disk rotate without slip. We need to find the equation of motion of the small disk. In otherwords, find the solution for θ'[t]

In[1]:=

```
Needs["Notation`"]
```

In[2]:=

```
Symbolize[m₁]
Symbolize[m₂]
Symbolize[I₁]
Symbolize[I₂]
```

Solve for ω to meet the no-slip condition

In[6]:=

```
Clear[ω, R, ϕ, θ, r, m₁, m₂, I₁, I₂, g]
noSlipEquation = (R - r) θ'[t] == R ϕ'[t] + r ω;
ω = ω /. First@Solve[noSlipEquation, ω]
```

Out[8]=

$$\frac{-r\,\theta'[t]+R\,\theta'[t]-R\,\phi'[t]}{r}$$

Find T and V and find the Lagrangian L=T-V

In[52]:=
$$T = \frac{1}{2} m_1 (R\,\phi'[t])^2 + \frac{1}{2} I_1\,\phi'[t]^2 + \frac{1}{2} m_2 \left((R\,\phi'[t] - (R - r)\,\theta'[t]\,\mathrm{Cos}[\theta[t]])^2 + ((R - r)\,\theta'[t]\,\mathrm{Sin}[\theta[t]])^2\right) + \frac{1}{2} I_2 \left(\omega^2\right)$$

Out[52]=
$$\frac{1}{2} I_1\,\phi'[t]^2 + \frac{1}{2} m_1 R^2\,\phi'[t]^2 + \frac{I_2\,(-r\,\theta'[t] + R\,\theta'[t] - R\,\phi'[t])^2}{2\,r^2} + \frac{1}{2} m_2 \left((-r + R)^2\,\mathrm{Sin}[\theta[t]]^2\,\theta'[t]^2 + (-(-r + R)\,\mathrm{Cos}[\theta[t]]\,\theta'[t] + R\,\phi'[t])^2\right)$$

In[53]:=
$$V = -m_2\,g\,(R - r)\,\mathrm{Cos}[\theta[t]]$$

Out[53]=
$$-g\,m_2\,(-r + R)\,\mathrm{Cos}[\theta[t]]$$

In[54]:=
```
L = (T - V) // FullSimplify
```

Out[54]=
$$\frac{1}{2}\left(2\,g\,m_2\,(-r + R)\,\mathrm{Cos}[\theta[t]] + \frac{1}{r^2}\left(\left(I_2 + m_2\,r^2\right)(r - R)^2\,\theta'[t]^2 + 2\,(r - R)\,R\left(I_2 + m_2\,r^2\,\mathrm{Cos}[\theta[t]]\right)\theta'[t]\,\phi'[t] + \left(I_1\,r^2 + \left(I_2 + (m_1 + m_2)\,r^2\right)R^2\right)\phi'[t]^2\right)\right)$$

Solve for ϕ''[t], note genearlized force is zero

In[55]:=
```
equationOfMotion1 = D[D[L, ϕ'[t]], t] - D[L, ϕ[t]] == 0 // Simplify
```

Out[55]=
$$\frac{1}{r^2}\left(-m_2\,r^2\,(r - R)\,R\,\mathrm{Sin}[\theta[t]]\,\theta'[t]^2 + (r - R)\,R\left(I_2 + m_2\,r^2\,\mathrm{Cos}[\theta[t]]\right)\theta''[t] + \left(I_1\,r^2 + \left(I_2 + (m_1 + m_2)\,r^2\right)R^2\right)\phi''[t]\right) == 0$$

Solve for θ''[t], note genearlized force is zero

In[56]:=
```
equationOfMotion2 = D[D[L, θ'[t]], t] - D[L, θ[t]] == 0 // Simplify
```

Out[56]=
$$\frac{1}{r}(r - R)\left(-g\,m_2\,r^2\,\mathrm{Sin}[\theta[t]] + \left(I_2 + m_2\,r^2\right)(r - R)\,\theta''[t] + R\left(I_2 + m_2\,r^2\,\mathrm{Cos}[\theta[t]]\right)\phi''[t]\right) == 0$$

Define problem parameters

In[14]:=

```
parameters = {g → 9.8, R -> 1, r -> .1, m1 -> 10, m2 -> 1};
parameters = Union[parameters, {I1 -> m1 R^2/2, I2 -> m2 r^2/2} /. parameters]
```

Out[15]=

```
{g → 9.8, I1 → 5, I2 → 0.005000000000000001, m1 → 10, m2 → 1, r → 0.1, R → 1}
```

Numerically solve the equations of motion using some initial conditions

In[64]:=

```
s = NDSolve[{equationOfMotion1, equationOfMotion2, ϕ[0] == 30 Degree,
    ϕ'[0] == -2, θ[0] == 0, θ'[0] == -2} /. parameters, {ϕ[t], θ[t]}, {t, 0, 30}]
```

Out[64]=

```
{{ϕ[t] → InterpolatingFunction[{{0., 30.}}, <>][t],
  θ[t] → InterpolatingFunction[{{0., 30.}}, <>][t]}}
```

Plot the solution, motion of small disk, we see the small disk will make an oscillation motion inside the large disk as the large disk is rotating

In[73]:=

```
Plot[Evaluate[θ[t] /. s], {t, 0, 30}, PlotRange → All, Frame → True,
 FrameLabel → {{"θ[t]", None}, {"time (sec)", "soution θ[t]"}},
 FrameTicks → {Automatic, {-30 Degree, -15 Degree, 15 Degree, 30 Degree}}]
```

Out[73]=